Mathematica Manual

Notebook 17: Integration in Vector Fields

■ Line Integrals

Example: Let $f(x, y, z) = \sqrt{1 + x^2 + y^2}$ and $\mathbf{r}(t) = t\mathbf{i} + t^2\mathbf{j} + t^3\mathbf{k}$, $0 \le t \le \frac{\pi}{2}$. Evaluate $\int_C f\, ds$ using the formulas in your text. You may change the terms in red for other functions.

First, the function f is defined along with the components of $\mathbf{r}(t)$. Then the magnitude of the velocity vector is found using the dot product.

```
In[1]:= Clear[f, x, y, z, g, h, k]
        f[x_, y_, z_] = √(1 + x^2 + y^2);
        {g[t_], h[t_], k[t_]} = {t, t^2, t^3};
        v[t_] = {g'[t], h'[t], k'[t]};
        dsdt = √(v[t].v[t])
```

Out[5]= $\sqrt{1 + 4\,t^2 + 9\,t^4}$

Next, the integrand is formed and simplified.

```
In[6]:= integrand = f[g[t], h[t], k[t]] dsdt // Simplify
```

Out[6]= $\sqrt{1 + t^2 + t^4}\,\sqrt{1 + 4\,t^2 + 9\,t^4}$

If you try to evaluate the integral with *Mathematica*, an answer is not found. Therefore the `NIntegrate` command is used instead to obtain a numerical answer.

In[7]:= $\int_0^{\frac{\pi}{2}} \text{integrand}\, dt$

Out[7]= $\int_0^{\frac{\pi}{2}} \sqrt{1 + t^2 + t^4}\,\sqrt{1 + 4\,t^2 + 9\,t^4}\, dt$

In[8]:= `NIntegrate[integrand, {t, 0, π/2}]`

Out[8]= 10.4518

■ Vector Fields, Work, Circulation, and Flux

Example: Find the work done by the force $\mathbf{F} = xz\,\mathbf{i} + z\mathbf{j} + yz\,\mathbf{k}$ over the curve $\mathbf{r}(t) = t^2\mathbf{i} + t\mathbf{j} + t^3\mathbf{k}$, $0 \le t \le 1$.

Study the following input and output used to solve the work problem. We are just defining functions here and there will be no output.

```
In[9]:= Clear[F, g, h, k, r, x, y, z, t]
        F[x_, y_, z_] = {x z, z, y z};
        {g[t_], h[t_], k[t_]} = {t^2, t, t^3};
        r[t_] = {g[t], h[t], k[t]};
```

We can look at the functions whose dot product we will be taking.

```
In[13]:= F[g[t], h[t], k[t]]
         r'[t]
```

Out[13]= $\{t^5, t^3, t^4\}$

Out[14]= $\{2\,t, 1, 3\,t^2\}$

In *Mathematica*, there is a command for the dot product that requires a special package, or, we can simply place a **period** between the two vectors that we are dotting into one another.

```
In[15]:= integrand = F[g[t], h[t], k[t]].r'[t]
```

Out[15]= $t^3 + 5\,t^6$

In[16]:= $\int_0^1 \text{integrand}\, dt$

Out[16]= $\frac{27}{28}$

■ Green's Theorem in the Plane

■ Circulation Around a Closed Path

Example: Find the counterclockwise circulation of the field $\mathbf{F} = (x - 2\,y)\mathbf{i} + (3\,x + y)\mathbf{j}$ around the simple closed curve C: $4\,x^2 + y^2 = 16$.

We will begin by defining the components of the force field and the curve. Then we will plot the vector field for the function together with the contour plot for the curve.

In[17]:=
```
Clear[x, y]
m[x_, y_] = x - 2 y; n[x_, y_] = 3 x + y;
g[x_, y_] = 4 x^2 + y^2
pv = VectorPlot[{m[x, y], n[x, y]},
   {x, -4, 4}, {y, -4, 4}, AxesLabel → {t, y}, VectorScale → {.1, .4}];
pc = ContourPlot[g[x, y], {x, -4, 4}, {y, -4, 4}, Contours → {16}, ContourShading → False];
Show[pv, pc]
```

Out[19]= $4 x^2 + y^2$

Out[22]=



Looking at this picture, you would expect to get a positive value for the circulation. The second form of Green's Theorem is used here to compute the circulation.

In[23]:=
```
integrand = ∂x n[x, y] - ∂y m[x, y]
```

$$\int_{-2}^{2}\int_{-\sqrt{16-4x^2}}^{\sqrt{16-4x^2}} \texttt{integrand}\, dy\, dx$$

Out[23]= 5

Out[24]= 40π

■ **Flux Across a Closed Region**

Example: Suppose we consider the same force field and closed curve as above, but compute the flux over the region instead of the circulation around the closed curve. This requires the first form of Green's theorem. Let us again look at the figure, but this time, picture what the flux might be.

In[25]:=
```
Clear[x, y]
m[x_, y_] = x - 2 y; n[x_, y_] = 3 x + y;
g[x_, y_] = 4 x^2 + y^2
pv = VectorPlot[{m[x, y], n[x, y]},
   {x, -4, 4}, {y, -4, 4}, AxesLabel → {t, y}, VectorScale → {.1, .4}];
pc = ContourPlot[g[x, y], {x, -4, 4}, {y, -4, 4}, Contours → {16}, ContourShading → False];
Show[pv, pc]
```

Out[27]= $4x^2 + y^2$

Out[30]=



Now, we need to determine the vector tangent to and then perpendicular to the closed curve. For ease in computing the upcoming line integral, we will write the closed curve in parametric form; you should note that this parametrization satisfies the constraint equation, g(x,y)=16.

In[31]:=
```
x = 2 Cos[t];
y = 4 Sin[t];
r = {x, y};
dr = D[r, t]
perp = {dr[[2]], -dr[[1]]}
```

Out[34]= {-2 Sin[t], 4 Cos[t]}

Out[35]= {4 Cos[t], 2 Sin[t]}

We will dot the vector perpendicular to the constraint into our force vector and integrate from 0 to 2π. Because we have defined x and y as functions of t, the m and n functions will be functions of t as well.

In[36]:=
```
lineintegrand = perp.{m[x, y], n[x, y]} // Simplify
```

$$\int_0^{2\pi} \text{lineintegrand}\, dt$$

Out[36]= 8 - 20 Cos[t] Sin[t]

Out[37]= 16π

We can verify this form of Green's Theorem by computing the following double integral. It is critical that we clear the assignments made for *x* and *y*, since, here, we are thinking of them as independent variables.

In[38]:=
```
Clear[x, y]
```
$$\int_{-2}^{2}\int_{-\sqrt{16-4x^2}}^{\sqrt{16-4x^2}} (D[m[x, y], x] + D[n[x, y], y])\, dy\, dx$$

Out[39]= 16π

Divergence and Surface Integrals

Using a Vector Package

We mentioned above that we could have loaded a special vector package to compute the dot product. However, since the period accomplishes the same task, we did not do that. Here, we will compute the cross product and the divergence, so we will load this special three-dimensional vector package to have a built-in way to perform these operations. We only need to load it once.

In[40]:=
```
Needs["VectorAnalysis`"]
```

We will begin by experimenting with some of the new commands that are now available to us.

In[41]:=
```
v1 = {2, 5, -3};
v2 = {-7, 1, 9};
v3 = {4, -6, 0};
DotProduct[v1, v2]
DotProduct[v1, v2, Cylindrical]
DotProduct[v1, v2, Spherical]
CrossProduct[v1, v2]
ScalarTripleProduct[v1, v2, v3]
```

Out[44]= -36

Out[45]= -27 - 14 Cos[1] Cos[5] - 14 Sin[1] Sin[5]

Out[46]= -14 Cos[1] Cos[5] - 14 Cos[3] Cos[9] Sin[1] Sin[5] + 14 Sin[1] Sin[3] Sin[5] Sin[9]

Out[47]= {48, 3, 37}

Out[48]= 174

In the VectorAnalysis package, the default coordinate system is the Cartesian system, but if you specify cylindrical or spherical, *Mathematica* assumes that the vectors given are in those coordinate systems.

Now we will use this package to compute the divergence of a force.

In[49]:=
```
Clear[x, y, z]
force = {x Cos[z], x^3 e^z, -y z};
Print["The divergence of this force is ", divf = Div[force, Cartesian[x, y, z]]]
```

```
The divergence of this force is -y + Cos[z]
```

Using Spherical Coordinates to Determine Flux Through a Surface

We will begin by drawing the force field and the sphere.

In[52]:=
```
Clear[x, y, z, u, v, t, r, θ, force]
force = {x Cos[z], x^3 ⅇ^z, -y z};
forceplot = VectorPlot3D[force, {x, -3, 3}, {y, -3, 3},
   {z, -3, 3}, VectorPoints → 8, VectorScale → {Small, Automatic, #3 &}];
sphereplot = SphericalPlot3D[3, {ϕ, 0, π}, {θ, 0, 2 π}, AxesLabel → {x, y, z}];
Show[forceplot, sphereplot]
```

Out[55]=



The surface integrals evaluating the flux across surfaces can require considerable set up. As you recall, parametrizations are very convenient in determining line integrals. Likewise, parametrizations can facilitate the computation of surface integrals. If we want to find the flux of the force defined above across a the sphere of radius 3, we can use spherical coordinates to write the equation of the sphere parametrically.

In[56]:=
```
Clear[x, y, z, r, θ, ϕ]
force := {x Cos[z], x^3 ⅇ^z, -y z};
x = 3 Sin[θ] Sin[ϕ];
y = 3 Cos[θ] Sin[ϕ];
z = 3 Cos[ϕ];
r = {x, y, z};
rtheta = D[r, θ];
rphi = D[r, ϕ];
cr = Cross[rtheta, rphi] // Simplify;
Print["The integrand for the spherical surface flux integral is ",
 integrand = force.cr // Simplify]
Print["The flux across the sphere is ", NIntegrate[integrand, {θ, 0, 2 π}, {ϕ, 0, π}]]
```

The integrand for the spherical surface flux integral is

$$27\,\mathrm{Sin}[\phi]^2\left(\mathrm{Cos}[3\,\mathrm{Cos}[\phi]]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi]-3\,\mathrm{Cos}[\theta]\left(\mathrm{Cos}[\phi]^2-3\,e^{3\,\mathrm{Cos}[\phi]}\,\mathrm{Sin}[\theta]^3\,\mathrm{Sin}[\phi]^3\right)\right)$$

The flux across the sphere is 39.0952