Mathematica Manual

Notebook 14: Vector-Valued Functions and Motion in Space

■ Vector Functions

This example begins with a position vector defined in parametric form.

Example: Let $\mathbf{r}(t) = [(4 + \sin(10\,t))\cos(t)]\mathbf{i} + [(4 + \sin(10\,t))\cos(t)]\,\mathbf{j} + (2\cos(10\,t))\mathbf{k}$ and complete each of the following.

(a) Plot the space curve traced out by the position vector for $0 \le t \le 2\pi$.

(b) Find the components of the velocity vector dr/dt.

(c) Evaluate the velocity vector at $t = \frac{\pi}{6}$ and determine the equation of the tangent line to the curve at $\mathbf{r}\left(\frac{\pi}{6}\right)$.

(d) Plot the tangent line together with the curve.

Part (a) First, the x, y and z components of the curve are defined parametrically and then the curve is plotted.

In[1]:=
```
x[t_] = (4 + Sin[10 t]) Cos[t];
y[t_] = (4 + Sin[10 t]) Sin[t];
z[t_] = 2 Cos[10 t];
spacecurve = ParametricPlot3D[{x[t], y[t], z[t]}, {t, 0, 2 π}, PlotPoints → 250, AxesLabel → {x, y,
```

Out[4]=



Part (b): Now the derivative of each component is computed to obtain the velocity vector.

In[5]:=
```
v[t_] = {x'[t], y'[t], z'[t]}
```

Out[5]=
```
{10 Cos[t] Cos[10 t] - Sin[t] (4 + Sin[10 t]),
 10 Cos[10 t] Sin[t] + Cos[t] (4 + Sin[10 t]), -20 Sin[10 t]}
```

Part (c) Next, the velocity vector is evaluated and the parametric equations of the tangent line are computed.

```
In[6]:= t0 = π/6;
        v[t0] // N
Out[7]= {2.76314, 5.2141, 17.3205}
```

We will write the equations of the tangent lines in parametric form.

```
In[8]:= xl[t_] := x[t0] + x'[t0] t;
        yl[t_] := y[t0] + y'[t0] t;
        zl[t_] := z[t0] + z'[t0] t;
```

Part (d) The space curve and the tangent line are displayed in the following output cell. Instead of plotting the curve for values smaller interval of values is used so that the tangent line is easier to see.

```
In[11]:= tanlineplot = ParametricPlot3D[{xl[t], yl[t], zl[t]}, {t, -π/4, π/4}];
         spacecurve = ParametricPlot3D[{x[t], y[t], z[t]}, {t, 0, π/4}, PlotPoints → 250, AxesLabel → {x, y, z
         Show[{spacecurve, tanlineplot}, PlotRange → {{2, 6}, {-3, 3}, {-2, 5}}]
```

Out[13]=



Following is an alternative approach; we simply define the curve directly using the vector **r** as a function of t, a, and b. Leave a spac and t and a in the Sine and Cosine expressions.

```
In[14]:= Clear[r, v, t, a, b, tanline]
         r[t_, a_, b_] := {Cos[a t], Sin[a t], b t}
         t0 = 3 π / 2; tmin = 0; tmax = 4 π;
         v[t_, a_, b_] = D[r[t, a, b], t]
         tanline[t_, a_, b_] = v[t0, a, b] t + r[t0, a, b]
```

Out[17]= $\{-a\,\mathrm{Sin}[a\,t],\ a\,\mathrm{Cos}[a\,t],\ b\}$

Out[18]= $\left\{\mathrm{Cos}\left[\frac{3\,a\,\pi}{2}\right]-a\,t\,\mathrm{Sin}\left[\frac{3\,a\,\pi}{2}\right],\ a\,t\,\mathrm{Cos}\left[\frac{3\,a\,\pi}{2}\right]+\mathrm{Sin}\left[\frac{3\,a\,\pi}{2}\right],\ \frac{3\,b\,\pi}{2}+b\,t\right\}$

Letting $b = 1$ and allowing a to vary with the **Manipulate** command, we get the following curves and tangent lines. We use the **Ev** facilitate the graphing process. Move the slider at the top to the right to see what happens as a increases

```
In[19]:= Manipulate[ParametricPlot3D[Evaluate[{r[t, a, 1], tanline[t, a, 1]}],
           {t, tmin, tmax}, AxesLabel → {x, y, z}, PlotRange → {-5, 10}], {a, 1, 6, .5}]
```



You may alter functions, domain points, and values of a and b as you wish by changing the terms in red.

■ Curvature and the Unit Normal Vector

Example: Let $\mathbf{r}(t) = (\cos t)\mathbf{i} + \left(2 - \frac{1}{2}\sin t\right)\mathbf{j}$, $0 \le t \le 2\pi$ and let $t_0 = \frac{\pi}{4}$. Complete each of the following steps.

(a) Plot the plane curve over the specified interval.

(b) Calculate the curvature κ of the curve at the given point t_0.

(c) Find the unit normal vector **N** at t_0.

(d) If **C**= a**i**+b**j** is the vector from the origin to the center (a, b) of the osculating circle, find the center **C** from the vector equation **C**=$\mathbf{r}(t_0) + \frac{1}{\kappa(t_0)}\mathbf{N}(t_0)$. The point $P(x_0, y_0)$ on the curve is given by the position vector $\mathbf{r}(t_0)$.

(e) Plot the curve and the osculating circle together.

Part (a) The curve is plotted parametrically with *Mathematica*.

```
In[20]:= Clear[x, y, t, n, c, κ]
         x[t_] = Cos[t]; y[t_] = 2 - Sin[t]/2;
         tmin = 0; tmax = 2 π;
         curve = ParametricPlot[{x[t], y[t]}, {t, tmin, tmax}, AspectRatio → 1]
```

Out[23]=



Part (b) Using the formula in your text, the value of κ is computed.

```
In[24]:= κ[t_] = Abs[x'[t] y''[t] - y'[t] x''[t]] / (x'[t]^2 + y'[t]^2)^(3/2);
         t0 = π/4
         Print["The curvature at ", t0, " is ", κ[t0] // N]
```

Out[25]= $\frac{\pi}{4}$

The curvature at $\frac{\pi}{4}$ is 1.01193

Part (c) Since N is a reserved *Mathematica* word, you can instead use n to represent the unit normal vector. We use a special for normal vector and then evaluate it at t0. Is this the formula you should always use to find the unit normal vector? Why or why not?

In[27]:=
```
n[t_] = {y'[t], -x'[t]} / Sqrt[x'[t]^2 + y'[t]^2];
n[t0]
```

Out[28]= $\left\{-\frac{1}{\sqrt{5}}, \frac{2}{\sqrt{5}}\right\}$

Part (d) The center of the osculating circle is found using the following input cell.

In[29]:=
```
r[t_] = {x[t], y[t]};
c = r[t0] + 1/κ[t0] n[t0] // N
```

Out[30]= {0.265165, 2.53033}

Part (e) The *Mathematica* command
`ContourPlot[f[x,y],{x, xmin, xmax},{y, ymin,  ymax},ContourShading->False, Contours->{c`
will plot the graph of $f(x, y) = c$. Therefore the following command is used to graph the osculating circle.

In[31]:=
```
osccir = ContourPlot[(x - c[[1]])^2 + (y - c[[2]])^2, {x, -1, 2}, {y, 1, 4}, ContourShading → False, Contour
```

Out[31]=



Now the curve and circle are shown together.

In[32]:=
```
Show[{curve, osccir}, PlotRange → {{-1, 1.6}, {1, 3.6}}, AspectRatio → 1]
```

Out[32]=



■ Torsion and the Unit Binormal Vector

Writing the position curve in parametric form and applying the formulas in your text, we get the following. Change only the term exercises. Some computations may take longer than others.

In[33]:=
```
Clear[r, v, a, t, speed]
mag[vector_] := √vector.vector
Print["The position vector is ", r[t_] = {ⅇ^t Sin[t], ⅇ^t Cos[t], ⅇ^t}]
Print["The velocity vector is ", v[t_] = r'[t]]
Print["The acceleration vector is ", a[t_] = v'[t]]
Print["The speed is ", speed[t_] = mag[v[t]]]
Print["The unit tangent vector is ", utan[t_] = v[t] / speed[t] // Simplify]
Print["The unit normal vector is ", un[t_] = Simplify[utan'[t]] / mag[utan'[t]] // Simplify]
Print["The unit binormal is ", ubn[t_] = Cross[utan[t], un[t]] // Simplify]
Print["The curvature is ", curvature[t_] = mag[Cross[v[t], a[t]]] / speed[t]^3 // Simplify]
Print["The torsion is ",
 torsion[t_] = Det[{v[t], a[t], a'[t]}] / (Cross[v[t], a[t]].Cross[v[t], a[t]]) // Simplify]
```

The position vector is $\{e^t \text{Sin}[t], e^t \text{Cos}[t], e^t\}$

The velocity vector is $\{e^t \text{Cos}[t] + e^t \text{Sin}[t], e^t \text{Cos}[t] - e^t \text{Sin}[t], e^t\}$

The acceleration vector is $\{2 e^t \text{Cos}[t], -2 e^t \text{Sin}[t], e^t\}$

The speed is $\sqrt{\left(e^{2t} + \left(e^t \text{Cos}[t] - e^t \text{Sin}[t]\right)^2 + \left(e^t \text{Cos}[t] + e^t \text{Sin}[t]\right)^2\right)}$

The unit tangent vector is $\left\{\frac{e^t (\text{Cos}[t] + \text{Sin}[t])}{\sqrt{3}\sqrt{e^{2t}}}, \frac{e^t (\text{Cos}[t] - \text{Sin}[t])}{\sqrt{3}\sqrt{e^{2t}}}, \frac{e^t}{\sqrt{3}\sqrt{e^{2t}}}\right\}$

The unit normal vector is $\left\{\frac{e^t (\text{Cos}[t] - \text{Sin}[t])}{\sqrt{2}\sqrt{e^{2t}}}, -\frac{e^t (\text{Cos}[t] + \text{Sin}[t])}{\sqrt{2}\sqrt{e^{2t}}}, 0\right\}$

The unit binormal is $\left\{\frac{\text{Cos}[t] + \text{Sin}[t]}{\sqrt{6}}, \frac{\text{Cos}[t] - \text{Sin}[t]}{\sqrt{6}}, -\sqrt{\frac{2}{3}}\right\}$

The curvature is $\frac{\sqrt{2}\sqrt{e^{4t}}}{3\left(e^{2t}\right)^{3/2}}$

The torsion is $-\frac{e^{-t}}{3}$

We can evaluate each of these functions at any point t0.

```
In[44]:= Print["t0 = ", t0 = Log[2]]
Print["At ", t0, ", the velocity = ", v[t0]]
Print["At ", t0, ". the acceleration = ", a[t0]]
Print["At ", t0, ", the speed = ", speed[t0]]
Print["At ", t0, ", the unit tangent = ", utan[t0]]
Print["At ", t0, ", the unit normal = ", un[t0]]
Print["At ", t0, ", the unit binormal = ", ubn[t0]]
Print["At ", t0, ", the curvature = ", curvature[t0]]
Print["At ", t0, ", the torsion = ", torsion[t0]]
```

t0 = Log[2]

At Log[2], the velocity = $\{2\,\text{Cos}[\text{Log}[2]] + 2\,\text{Sin}[\text{Log}[2]],\ 2\,\text{Cos}[\text{Log}[2]] - 2\,\text{Sin}[\text{Log}[2]],\ 2\}$

At Log[2]. the acceleration = $\{4\,\text{Cos}[\text{Log}[2]],\ -4\,\text{Sin}[\text{Log}[2]],\ 2\}$

At Log[2], the speed = $\sqrt{\left(4 + (2\,\text{Cos}[\text{Log}[2]] - 2\,\text{Sin}[\text{Log}[2]])^2 + (2\,\text{Cos}[\text{Log}[2]] + 2\,\text{Sin}[\text{Log}[2]])^2\right)}$

At Log[2], the unit tangent = $\left\{\frac{\text{Cos}[\text{Log}[2]] + \text{Sin}[\text{Log}[2]]}{\sqrt{3}},\ \frac{\text{Cos}[\text{Log}[2]] - \text{Sin}[\text{Log}[2]]}{\sqrt{3}},\ \frac{1}{\sqrt{3}}\right\}$

At Log[2], the unit normal = $\left\{\frac{\text{Cos}[\text{Log}[2]] - \text{Sin}[\text{Log}[2]]}{\sqrt{2}},\ -\frac{\text{Cos}[\text{Log}[2]] + \text{Sin}[\text{Log}[2]]}{\sqrt{2}},\ 0\right\}$

At Log[2], the unit binormal = $\left\{\frac{\text{Cos}[\text{Log}[2]] + \text{Sin}[\text{Log}[2]]}{\sqrt{6}},\ \frac{\text{Cos}[\text{Log}[2]] - \text{Sin}[\text{Log}[2]]}{\sqrt{6}},\ -\sqrt{\frac{2}{3}}\right\}$

At Log[2], the curvature = $\frac{1}{3\sqrt{2}}$

At Log[2], the torsion = $-\frac{1}{6}$

If we want to see the decimal approximations to the above values, we can simply add a decimal point after the 2.

In[53]:=
```
Print["t0 = ", t0 = Log[2.]]
Print["At ", t0, ", the velocity = ", v[t0]]
Print["At ", t0, ". the acceleration = ", a[t0]]
Print["At ", t0, ", the speed = ", speed[t0]]
Print["At ", t0, ", the unit tangent = ", utan[t0]]
Print["At ", t0, ", the unit normal = ", un[t0]]
Print["At ", t0, ", the unit binormal = ", ubn[t0]]
Print["At ", t0, ", ", "the curvature = ", curvature[t0]]
Print["At ", t0, ", ", "the torsion = ", torsion[t0]]
```

t0 = 0.693147

At 0.693147, the velocity = {2.8164, 0.260555, 2.}

At 0.693147. the acceleration = {3.07696, −2.55585, 2.}

At 0.693147, the speed = 3.4641

At 0.693147, the unit tangent = {0.813025, 0.0752158, 0.57735}

At 0.693147, the unit normal = {0.0921202, −0.995748, 0}

At 0.693147, the unit binormal = $\left\{0.574895,\ 0.0531856,\ -\sqrt{\frac{2}{3}}\right\}$

At 0.693147, the curvature = 0.235702

At 0.693147, the torsion = −0.166667